
df1gw update 10.2001

labels:
a Arzneitaxe 100
d DEM > 560
f  > DEM 510
k Kalkulation 16% 10
l Kalkulation 7% 400
m Memory 600
n Naturalrabatt 300
z PZN 800

subrout:
20 Kalkulationsformat
50 Kalkulation
900 Rundung

mem:
d Eingabe rechnen
e Eingabe ohne Änderung
k  Faktor
o zwischen
p Eingabe Prozent
w Steuer

pass "df1gw"
 
5 rem datentypist
10 "k" aread d
15 w=.16
20 using "#####.##"
22 print "ek dm? ";d
40 input "kalk% ? ";p
50 d=d+d*p/100
55 gosub 900
60 print "    dm ";d
62 m=d
64 d=w*d
66 gosub 900
70 print "  mwst ";d
72 d=d+m
80 print "**vk** ";d
90 end
 
100     "a" aread d
110     using "#####.##"
115     w=.16
120     print "ek dm ";d
130     if d<2.41 then let p=68: goto 50
140     if d<2.64 then let d=d+1.63: goto 60
150     if d<7.61 then let p=62: goto 50
160     if d<8.27 then let d=d+4.71: goto 60
170     if d<14.29 then let p=57: goto 50
180     if d<16.97 then let d=d+8.14: goto 60
190     if d<23.76 then let p=48: goto 50
200     if d<26.52 then let d=d+11.4: goto 60
210     if d<38.01 then let p=43: goto 50
220     if d<44.17 then let d=d+16.34: goto 60
230     if d<57.01 then let p=37: goto 50
240     if d<70.31 then let d=d+21.09: goto 60



250     if d<1063.82 then let p=30: goto 50
260     print "Kappung"
270     end : rem Kappung nicht prgrammiert
 
300     "n" aread d
305     using "####"
310     input "natrab? ";e
320     print d;"+";e;"=";e*100/(d+e);"%"
330     end
 
400     "l" aread d
410     w=.07
420     goto 20

510 "f" aread d
520 k=1.95583
530 d=d*k
540 gosub 900
545 print d
550 end
 
560 "d" aread d
565 k=1.95583
570 d=d/k
575 gosub 900
577 print d
580 end
 
600     "m"
601     using
608     print mem ;"fr*";1278- mem ;"pgm"
616     print "df1gw"
620     end
 
800     "z" aread d
805     using
810     o=0:e=d: if d>999999 then end
820     for n=6 to 1 step -1
830     d=d/10^(n-1):m= int d
840     d=d-m:d=d*10^(n-1)
850     o=o+m*(8-n)
860     next n
870     p=o:o=( int (o/11))*11:p=p-o
880     print "pzn ";e;" ";p
890     end
900     d=d*100
910     if (d- int (d))<.5 goto 950
920     d=( int (d)+1)/100
930     return
950     d= int (d)/100
960     return


